In[3]:= `NotebookDirectory[]`

Out[3]= `C:\Dropbox\Work\myweb\Courses\Math_pages\Math_430\`

Examples of Fourier series

Preliminaries

Below is the definition of a periodic extension of a function defined on $(-L, L]$. This definition takes a function as a variable. The function has to be inputted as a so called pure function (that is instead of the variable we put # and the formula ends with &).

In[4]:=

```
Clear[ff, fft, x, lL, LL];
fft[ff_, x_, LL_] := ff[x - (Ceiling[(x - (LL))/(2 LL)]) (2 LL)]
```

In[6]:= `(#^2) &[2]`

Out[6]= 4

In[7]:= `fft[#^2 &, x, 1]`

Out[7]= $\left(x - 2\,\text{Ceiling}\left[\frac{1}{2}(-1+x)\right]\right)^2$

In[8]:= `Plot[fft[#^2 &, x, 1], {x, -4, 4}]`

Out[8]=



Example -1

In[9]:=
```
Clear[can1, cbn1, ffn1, n, lL, nn];

ffn1[x_] = Sign[x];
```

In[11]:=
```
cbn1[n_, lL_] = FullSimplify[
  1/LL Integrate[ffn1[x] Sin[n Pi/LL x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```

Out[11]= $-\frac{2\left(-1+(-1)^{n}\right)}{n\pi}$

In[12]:=
```
can1[0, LL_] = FullSimplify[1/(2 LL) Integrate[ffn1[x], {x, -LL, LL}], And[LL > 0]]
```

Out[12]= 0

In[13]:=
```
can1[n_, LL_] = FullSimplify[
  1/LL Integrate[ffn1[x] Cos[n Pi/LL x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```

Out[13]= 0

In[14]:=
```
nn = 10;
can1[0, lL] + Sum[can1[n, lL] Cos[n Pi/lL x], {n, 1, nn}] +
 Sum[cbn1[n, lL] Sin[n Pi/lL x], {n, 1, nn}]
```

Out[14]= $\frac{4\,\mathrm{Sin}\left[\frac{\pi x}{lL}\right]}{\pi}+\frac{4\,\mathrm{Sin}\left[\frac{3\pi x}{lL}\right]}{3\pi}+\frac{4\,\mathrm{Sin}\left[\frac{5\pi x}{lL}\right]}{5\pi}+\frac{4\,\mathrm{Sin}\left[\frac{7\pi x}{lL}\right]}{7\pi}+\frac{4\,\mathrm{Sin}\left[\frac{9\pi x}{lL}\right]}{9\pi}$

In[15]:=
```
Module[{pic1, pic2, pic2a, pic3, lL, nn}, nn = 20;
 lL = 1;
 pic1 = Plot[{ffn1[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.01], Blue}}];
 pic2 = Plot[{fft[ffn1[#] &, x, lL]}, {x, -5, 10},
   PlotStyle → {{Thickness[0.005], Green}}, Exclusions → Range[-10, 14, 1]];

 pic2a = Graphics[{
     {PointSize[0.015], Green, {Point[{# lL, -1}], Point[{# lL, 1}], Point[{# lL, 0}]} & /@
       Range[-10, 13, 1]}, {PointSize[0.01], White,
      {Point[{# lL, -1}], Point[{# lL, 1}]} & /@ Range[-10, 13, 1]}
   }];
 pic3 = Plot[Evaluate[{can1[0, lL] + Sum[can1[n, lL] Cos[(n Pi)/lL x], {n, 1, nn}] +
      Sum[cbn1[n, lL] Sin[(n Pi)/lL x], {n, 1, nn}]}], {x, -12, 14},
   PlotStyle → {{Thickness[0.003], Black}}, PlotRange → {{-4, 7}, {-1.5, 1.5}}];
 Show[pic1, pic2, pic2a, pic3, PlotRange → {{-3, 5}, {-1.5, 1.5}},
  AspectRatio → Automatic, ImageSize → 600]]
```

Out[15]=



The Fourier series of the above function is with L = 1

$$\frac{4}{\pi} \sum_{k=1}^{\infty} \frac{1}{2k-1} \text{Sin}[(2k-1)\pi x]$$

It converges pointwise to the Fourier periodic extension of Sign[x] with period 2.

In[16]:=
```
Module[{pic1, pic2, pic2a, pic3, lL, nn}, nn = 20;
 lL = 1;
 pic1 = Plot[{ffn1[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.007], Blue}}];
 pic2 = Plot[{fft[ffn1[#] &, x, lL]}, {x, -5, 10},
   PlotStyle → {{Thickness[0.005], Green}}, Exclusions → Range[-10, 14, 1]];

 pic2a = Graphics[{
    {PointSize[0.015], Green, {Point[{# lL, -1}], Point[{# lL, 1}], Point[{# lL, 0}]} & /@
      Range[-10, 13, 1]}, {PointSize[0.01], White,
     {Point[{# lL, -1}], Point[{# lL, 1}]} & /@ Range[-10, 13, 1]}
   }];
 pic3 = Plot[Evaluate[{can1[0, lL] + Sum[can1[n, lL] Cos[(n Pi)/lL x], {n, 1, nn}] +
      Sum[cbn1[n, lL] Sin[(n Pi)/lL x], {n, 1, nn}]}], {x, -12, 14},
   PlotStyle → {{Thickness[0.001], Black}}, PlotRange → {{-4, 7}, {-1.5, 1.5}}];
 Show[pic1, pic2, pic2a, pic3, PlotRange → {{-3, 5}, {-1.5, 1.5}},
  AspectRatio → Automatic, ImageSize → 600]]
```

Out[16]=



Or, the same picture with Manipulate

In[17]:=
```
Manipulate[Module[{pic1, pic2, pic2a, pic3, lL}, lL = 1;
  pic1 = Plot[{ffn1[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.01], Blue}}];
  pic2 = Plot[{fft[ffn1[#] &, x, lL]}, {x, -5, 10},
    PlotStyle → {{Thickness[0.005], Green}}, Exclusions → Range[-10, 14, 1]];

  pic2a = Graphics[{
     {PointSize[0.015], Green, {Point[{# lL, -1}], Point[{# lL, 1}], Point[{# lL, 0}]} & /@
       Range[-10, 13, 1]}, {PointSize[0.01], White,
      {Point[{# lL, -1}], Point[{# lL, 1}]} & /@ Range[-10, 13, 1]}
    }];
  pic3 = Plot[Evaluate[{can1[0, lL] + Sum[can1[n, lL] Cos[(n Pi)/lL x], {n, 1, nn}] +
       Sum[cbn1[n, lL] Sin[(n Pi)/lL x], {n, 1, nn}]}], {x, -12, 14},
    PlotStyle → {{Thickness[0.003], Black}}, PlotRange → {{-4, 7}, {-1.5, 1.5}}];
  Show[pic1, pic2, pic2a, pic3, PlotRange → {{-3, 5}, {-1.5, 1.5}},
   AspectRatio → Automatic, ImageSize → 600]],
 {{nn, 10, "n"}, Join[Range[10], Range[15, 30, 5]]}, ControlType → Setter]
```

Out[17]=



What is important to point out here, is that for a specific x from the convergence theorem we KNOW the sum of this numerical series. For example for x=1/2

$$\frac{4}{\pi}\sum_{k=1}^{\infty}\frac{1}{2k-1}\,\mathrm{Sin}\left[(2k-1)\,\pi\,\frac{1}{2}\right]$$

or

$$\frac{4}{\pi}\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{2k-1}=1$$

Mathematica knows this

In[18]:= $\frac{4}{\pi}\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{2k-1}$

Out[18]= 1

In other words

$$\sum_{k=1}^{\infty}\frac{(-1)^{k-1}}{2k-1}=\frac{\pi}{4}$$

which is the famous Leibniz formula for π.

But we can get more numerical series sums from the above Fourier series. For x=1/3, the sum is also 1

In[19]:= $\frac{4}{\pi}\sum_{k=1}^{\infty}\frac{1}{2k-1}\,\mathrm{Sin}\left[(2k-1)*\pi*\frac{1}{3}\right]$

Out[19]= $\frac{2\left(\mathrm{ArcTan}\left[(-1)^{1/6}\right]-i\,\mathrm{ArcTanh}\left[(-1)^{1/3}\right]\right)}{\pi}$

Mathematica knows this.

In[20]:= $\mathrm{FullSimplify}\left[\frac{4}{\pi}\sum_{k=1}^{\infty}\frac{1}{2k-1}\,\mathrm{Sin}\left[(2k-1)*\frac{\mathrm{Pi}}{3}\right]\right]$

Out[20]= 1

But which numerical series is this?

In[21]:= $\mathrm{Table}\left[\left\{k,\frac{1}{2k-1}\,\mathrm{Sin}\left[(2k-1)*\frac{\mathrm{Pi}}{3}\right]\right\},\{k,1,20\}\right]$

Out[21]= $\left\{\left\{1,\frac{\sqrt{3}}{2}\right\},\{2,0\},\left\{3,-\frac{\sqrt{3}}{10}\right\},\left\{4,\frac{\sqrt{3}}{14}\right\},\{5,0\},\left\{6,-\frac{\sqrt{3}}{22}\right\},\left\{7,\frac{\sqrt{3}}{26}\right\},\{8,0\},\left\{9,-\frac{\sqrt{3}}{34}\right\},\left\{10,\frac{\sqrt{3}}{38}\right\},\{11,0\},\left\{12,-\frac{\sqrt{3}}{46}\right\},\left\{13,\frac{\sqrt{3}}{50}\right\},\{14,0\},\left\{15,-\frac{\sqrt{3}}{58}\right\},\left\{16,\frac{\sqrt{3}}{62}\right\},\{17,0\},\left\{18,-\frac{\sqrt{3}}{70}\right\},\left\{19,\frac{\sqrt{3}}{74}\right\},\{20,0\}\right\}$

We can factor out $\frac{\sqrt{3}}{2}$ and the the nonzero terms are

In[22]:= $\mathrm{Table}\left[\left\{j,3\,\mathrm{Floor}[j/2]+\frac{1+(-1)^{j-1}}{2}\right\},\{j,1,20\}\right]$

Out[22]= {{1, 1}, {2, 3}, {3, 4}, {4, 6}, {5, 7}, {6, 9}, {7, 10}, {8, 12}, {9, 13}, {10, 15}, {11, 16}, {12, 18}, {13, 19}, {14, 21}, {15, 22}, {16, 24}, {17, 25}, {18, 27}, {19, 28}, {20, 30}}

And the numerators are

In[23]:= $\text{Table}\left[\left\{j,\ 2\left(3\,\text{Floor}[j/2] + \frac{1+(-1)^{j-1}}{2}\right) - 1\right\},\ \{j, 1, 10\}\right]$

Out[23]= {{1, 1}, {2, 5}, {3, 7}, {4, 11}, {5, 13}, {6, 17}, {7, 19}, {8, 23}, {9, 25}, {10, 29}}

or

In[24]:= $\text{Table}\left[\left\{j,\ 3\left(j - \frac{1+(-1)^{j-1}}{2}\right) + (-1)^{j-1}\right\},\ \{j, 1, 10\}\right]$

Out[24]= {{1, 1}, {2, 5}, {3, 7}, {4, 11}, {5, 13}, {6, 17}, {7, 19}, {8, 23}, {9, 25}, {10, 29}}

or

In[25]:= $\text{Table}\left[\left\{j,\ \frac{6j - 3 - (-1)^{j-1}}{2}\right\},\ \{j, 1, 10\}\right]$

Out[25]= {{1, 1}, {2, 5}, {3, 7}, {4, 11}, {5, 13}, {6, 17}, {7, 19}, {8, 23}, {9, 25}, {10, 29}}

And the signs are

In[26]:= $\text{Table}\left[\left\{j,\ 3\,\text{Floor}[j/2] + \frac{1+(-1)^{j-1}}{2},\ (-1)^{j-1}\right\},\ \{j, 1, 10\}\right]$

Out[26]= {{1, 1, 1}, {2, 3, -1}, {3, 4, 1}, {4, 6, -1}, {5, 7, 1}, {6, 9, -1}, {7, 10, 1}, {8, 12, -1}, {9, 13, 1}, {10, 15, -1}}

Verify:

In[27]:= $\text{Table}\left[\left\{3\,\text{Floor}[j/2] + \frac{1+(-1)^{j-1}}{2},\ \frac{2}{\sqrt{3}}\left(\frac{1}{2k-1}\text{Sin}\left[(2k-1) * \frac{\text{Pi}}{3}\right]\right) /.\ \left\{k \to 3\,\text{Floor}[j/2] + \frac{1+(-1)^{j-1}}{2}\right\},\ \frac{2(-1)^{j-1}}{6j - 3 + (-1)^{j}}\right\},\ \{j, 1, 10\}\right]$

Out[27]= $\left\{\{1, 1, 1\}, \left\{3, -\frac{1}{5}, -\frac{1}{5}\right\}, \left\{4, \frac{1}{7}, \frac{1}{7}\right\}, \left\{6, -\frac{1}{11}, -\frac{1}{11}\right\}, \left\{7, \frac{1}{13}, \frac{1}{13}\right\}, \left\{9, -\frac{1}{17}, -\frac{1}{17}\right\}, \left\{10, \frac{1}{19}, \frac{1}{19}\right\}, \left\{12, -\frac{1}{23}, -\frac{1}{23}\right\}, \left\{13, \frac{1}{25}, \frac{1}{25}\right\}, \left\{15, -\frac{1}{29}, -\frac{1}{29}\right\}\right\}$

Next command verifies the first 10000 terms:

In[28]:= $\text{Apply}\left[\text{And},\ \text{Table}\left[\left(\frac{2}{\sqrt{3}}\left(\frac{1}{2k-1}\text{Sin}\left[(2k-1) * \frac{\text{Pi}}{3}\right]\right) /.\ \left\{k \to 3\,\text{Floor}[j/2] + \frac{1+(-1)^{j-1}}{2}\right\}\right) == \frac{2(-1)^{j-1}}{6j - 3 + (-1)^{j}},\ \{j, 1, 10000\}\right]\right]$

Out[28]= True

So, the sum of this series should be 1

In[29]:= FullSimplify$\left[\frac{4\sqrt{3}}{\pi}\sum_{j=1}^{\infty}\frac{(-1)^{j-1}}{6j-3+(-1)^{j}}\right]$

Out[29]= $\frac{4\sqrt{3}\sum_{j=1}^{\infty}\frac{(-1)^{-1+j}}{-3+(-1)^{j}+6j}}{\pi}$

So, Mathematica does not know the sum of this series. We can verify numerically

In[30]:= N$\left[\frac{4\sqrt{3}}{\pi}\sum_{j=1}^{100000}\frac{(-1)^{j-1}}{6j-3+(-1)^{j}}\right]$

Out[30]= 0.999998

Numerical evidence shows that it is correct.

Example 0

```
In[31]:= Clear[ca0, cb0, ff0, n, lL, nn];

         ff0[x_] = UnitStep[x];
```

In[33]:= cb0[n_, LL_] = FullSimplify$\left[\frac{1}{LL}\right.$ Integrate$\left[\text{ff0}[x]\ \text{Sin}\left[\frac{n\,Pi}{LL}x\right], \{x, -LL, LL\}\right]$, And[LL > 0, n ∈ Integers, n > 0]$\left.\right]$

Out[33]= $\frac{1+(-1)^{1+n}}{n\pi}$

In[34]:= ca0[0, LL_] = FullSimplify$\left[\frac{1}{2\,LL}\right.$ Integrate[ff0[x], {x, -LL, LL}], And[LL > 0]$\left.\right]$

Out[34]= $\frac{1}{2}$

In[35]:= ca0[0, 3]

Out[35]= $\frac{1}{2}$

In[36]:= ca0[n_, LL_] = FullSimplify$\left[\frac{1}{LL}\right.$ Integrate$\left[\text{ff0}[x]\ \text{Cos}\left[\frac{n\,Pi}{LL}x\right], \{x, -LL, LL\}\right]$, And[LL > 0, n ∈ Integers, n > 0]$\left.\right]$

Out[36]= 0

In[37]:=
```
Module[{pic1, pic2, pic2a, pic3, lL, nn}, lL = 1;
 nn = 10;
 pic1 = Plot[{ff0[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.01], Blue}}];
 pic2 = Plot[{fft[ff0[#] &, x, lL]}, {x, -5, 10},
   PlotStyle → {{Thickness[0.005], Green}}, Exclusions → Range[-10, 14, 1]];

 pic2a = Graphics[{
    {PointSize[0.02], Green,
     {Point[{# lL, 0}], Point[{# lL, 1}], Point[{# lL, 1 / 2}]} & /@ Range[-10, 13, 1]},
    {PointSize[0.014], White, {Point[{# lL, 0}], Point[{# lL, 1}]} & /@ Range[-10, 13, 1]}
   }];
 pic3 = Plot[Evaluate[ca0[0, lL] + Sum[ca0[n, lL] Cos[(n Pi)/lL x], {n, 1, nn}] + Sum[cb0[n, lL]
       Sin[(n Pi)/lL x], {n, 1, nn}]], {x, -12, 14}, PlotStyle → {{Thickness[0.003], Black}}];
 Show[pic1, pic2, pic2a, pic3, PlotRange → {{-2, 3}, {-.5, 1.5}},
  AspectRatio → Automatic, ImageSize → 600]]
```

Out[37]=



Example 1

In[38]:=
```
Clear[ca1, cb1, ff1, n, lL, nn];

ff1[x_] = x;
```

In[40]:=
```
ca1[n_, LL_] = FullSimplify[
  1/LL Integrate[ff1[x] Cos[(n Pi)/LL x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```

Out[40]= 0

In[41]:=
```
ca1[0, LL_] =
  FullSimplify[1/(2 LL) Integrate[ff1[x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```

Out[41]= 0

In[42]:=
```
cb1[n_, LL_] = FullSimplify[
  1/LL Integrate[ff1[x] Sin[(n Pi)/LL x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```

Out[42]= $-\frac{2\ (-1)^{n}\ \mathrm{LL}}{n\ \pi}$

In[43]:=
```
Module[{pic1, pic2, pic2a, pic3, nn, lL}, lL = 2; nn = 10;
 pic1 = Plot[{ff1[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.01], Blue}}];
 pic2 = Plot[{fft[ff1[#] &, x, lL]}, {x, -5, 10},
   PlotStyle → {{Thickness[0.005], Green}}, Exclusions → Range[-10, 14, 4]];

 pic2a = Graphics[{
    {PointSize[0.02], Green,
     {Point[{#, -2}], Point[{#, 2}], Point[{#, 0}]} & /@ Range[-10, 13, 4]},
    {PointSize[0.014], White, {Point[{#, -2}], Point[{#, 2}]} & /@ Range[-10, 13, 4]}
   }];
 pic3 = Plot[Evaluate[Sum[cb1[n, lL] Sin[(n Pi)/lL x], {n, 1, nn}]],
   {x, -12, 14}, PlotStyle → {{Thickness[0.003], Black}}];
 Show[pic1, pic2, pic2a, pic3, PlotRange → {{-4, 11}, {-3, 3}},
  AspectRatio → Automatic, ImageSize → 800]]
```

Out[43]=



Example 2

In[44]:=
```
Clear[ca2, cb2, ff2, n, lL, nn];

ff2[x_] = Abs[x];
```

In[46]:=
```
cb2[n_, LL_] = FullSimplify[
   1/LL Integrate[ff2[x] Sin[(n Pi)/LL x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```
Out[46]= 0

In[47]:=
```
ca2[0, LL_] = FullSimplify[1/(2 LL) Integrate[ff2[x], {x, -LL, LL}], And[LL > 0]]
```
Out[47]= $\frac{LL}{2}$

In[48]:=
```
ca2[n_, LL_] = FullSimplify[
   1/LL Integrate[ff2[x] Cos[(n Pi)/LL x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```
Out[48]= $\frac{2\left(-1+(-1)^n\right) LL}{n^2 \pi^2}$

In[49]:=
```
Module[{pic1, pic2, pic2a, pic3, nn, lL}, lL = 2;
 nn = 10;
 pic1 = Plot[{ff2[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.007], Blue}}];
 pic2 = Plot[{fft[ff2[#] &, x, lL]}, {x, -5, 10},
   PlotStyle → {{Thickness[0.005], Green}}, Exclusions → Range[-10, 14, 4]];
 pic3 = Plot[Evaluate[{ca2[0, lL] + Sum[ca2[n, lL] Cos[(n Pi)/lL x], {n, 1, nn}] +
       Sum[cb2[n, lL] Sin[(n Pi)/lL x], {n, 1, nn}]}],
   {x, -12, 14}, PlotStyle → {{Thickness[0.002], Black}}];
 Show[pic1, pic2, pic3, PlotRange → {{-4, 11}, {-1, 3}},
  AspectRatio → Automatic, ImageSize → 800]]
```

Out[49]=


We can clearly see the uniform convergence here.

One interesting numerical series when we substitute x = L in the Fourier Series for this function:

$$\frac{L}{2}+\frac{2L}{\pi^2}\sum_{n=1}^{\infty}\frac{(-1+(-1)^n)}{n^2}\text{Cos}\left[\frac{n\pi}{L}x\right] = \frac{L}{2}-\frac{4L}{\pi^2}\sum_{k=1}^{\infty}\frac{1}{(2k-1)^2}\text{Cos}\left[\frac{(2k-1)\pi}{L}x\right]$$

$$L = \frac{L}{2}+\frac{2L}{\pi^2}\sum_{n=1}^{\infty}\frac{(-1+(-1)^n)}{n^2}\text{Cos}[n\pi] =$$

$$\frac{L}{2}-\frac{4L}{\pi^2}\sum_{k=1}^{\infty}\frac{1}{(2k-1)^2}\text{Cos}[(2k-1)\pi] = \frac{L}{2}+\frac{4L}{\pi^2}\sum_{k=1}^{\infty}\frac{1}{(2k-1)^2}$$

Therefore

$$\frac{1}{2} = \frac{4}{\pi^2}\sum_{k=1}^{\infty}\frac{1}{(2k-1)^2}$$

$$\frac{\pi^2}{8} = \sum_{k=1}^{\infty}\frac{1}{(2k-1)^2}$$

And from here we can calculate

$$S = \sum_{n=1}^{\infty}\frac{1}{n^2} = \sum_{k=1}^{\infty}\frac{1}{(2k-1)^2}+\sum_{k=1}^{\infty}\frac{1}{(2k)^2} = \sum_{k=1}^{\infty}\frac{1}{(2k-1)^2}+\frac{1}{4}\sum_{k=1}^{\infty}\frac{1}{k^2} = \frac{\pi^2}{8}+\frac{1}{4}S$$

And from here one gets

$$\frac{\pi^2}{6} = S = \sum_{n=1}^{\infty}\frac{1}{n^2}$$

Example 3

In[50]:=

```
Clear[ff3, n, LL, lL, nn];

ff3[x_] = x UnitStep[x];
```

For this function we do not need to calculate the Fourier Coefficients. We know that they are (1/2) of the coefficients in E2 and E1.

In[52]:=
```
Module[{pic1, pic2, pic2a, pic3, nn, lL}, nn = 20;
  lL = 2;
  pic1 = Plot[{ff3[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.007], Blue}}];
  pic2 = Plot[{fft[ff3[#] &, x, lL]}, {x, -5, 10},
    PlotStyle → {{Thickness[0.004], Green}}, Exclusions → Range[-10, 14, 4]];

  pic2a = Graphics[{
     {PointSize[0.015], Green,
      {Point[{#, 0}], Point[{#, 2}], Point[{#, 1}]} & /@ Range[-10, 13, 4]},
     {PointSize[0.01], White, {Point[{#, 0}], Point[{#, 2}]} & /@ Range[-10, 13, 4]}
    }];
  pic3 = Plot[Evaluate[{1/2 ca2[0, lL] + Sum[1/2 ca2[n, lL] Cos[n Pi/lL x], {n, 1, nn}] +
       Sum[1/2 cb1[n, lL] Sin[n Pi/lL x], {n, 1, nn}]}],
    {x, -12, 14}, PlotStyle → {{Thickness[0.00175], Black}}];
  Show[pic1, pic2, pic2a, pic3, PlotRange → {{-3, 7}, {-1, 3}},
   AspectRatio → Automatic, ImageSize → 800]]
```

Out[52]=



Example 4

In[53]:=

```
Clear[ca4, cb4, ff4, n, lL, nn];

ff4[x_] = x^2 UnitStep[x];
```

In[55]:=

```
cb4[n_, LL_] = FullSimplify[
  1/LL Integrate[ff4[x] Sin[n Pi/LL x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```

Out[55]= $-\frac{LL^2 \left(2+(-1)^n \left(-2+n^2 \pi^2\right)\right)}{n^3 \pi^3}$

In[56]:=

```
ca4[0, LL_] = FullSimplify[1/(2 LL) Integrate[ff4[x], {x, -LL, LL}], And[LL > 0]]
```

Out[56]= $\frac{LL^2}{6}$

In[57]:=

```
ca4[n_, LL_] = FullSimplify[
  1/LL Integrate[ff4[x] Cos[n Pi/LL x], {x, -LL, LL}], And[LL > 0, n ∈ Integers, n > 0]]
```

Out[57]= $\frac{2 (-1)^n LL^2}{n^2 \pi^2}$

In[58]:=
```
Module[{pic1, pic2, pic2a, pic3, nn, lL}, lL = 1;
 nn = 20;
 pic1 = Plot[{ff4[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.008], Blue}}];

 pic2 = Plot[{fft[ff4[#] &, x, lL]}, {x, -5, 10},
   PlotStyle → {{Thickness[0.004], Green}}, Exclusions → Range[-11, 14, 2]];

 pic2a = Graphics[{
    {PointSize[0.01], Green,
     {Point[{#, 0}], Point[{#, 1}], Point[{#, 1 / 2}]} & /@ Range[-11, 13, 2]},
    {PointSize[0.007], White, {Point[{#, 0}], Point[{#, 1}]} & /@ Range[-11, 13, 2]}
   }];
 pic3 = Plot[Evaluate[{ca4[0, lL] + Sum[ca4[n, lL] Cos[n Pi/lL x], {n, 1, nn}] +
      Sum[cb4[n, lL] Sin[n Pi/lL x], {n, 1, nn}]}],
   {x, -12, 14}, PlotStyle → {{Thickness[0.002], Black}}];
 Show[pic1, pic2, pic2a, pic3, PlotRange → {{-3, 5}, {-.1, 1.1}},
  AspectRatio → Automatic, ImageSize → 800]]
```

Out[58]=



Example 5

In[59]:=

```
Clear[ca5, cb5, ff5, n, lL, nn];

ff5[x_] = x^2;
```

Since we already calculated E4 we do not need to calculate the Fourier coefficients for this function. We know that the sine coefficients are 0s and the cosine coefficients are double the cosine coefficients from E4.

In[61]:=
```
Module[{pic1, pic2, pic2a, pic3, nn, lL}, lL = 1;
 nn = 10;
 pic1 = Plot[{ff5[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.007], Blue}}];

 pic2 = Plot[{fft[ff5[#] &, x, lL]}, {x, -5, 10},
   PlotStyle → {{Thickness[0.005], Green}}, Exclusions → Range[-11, 14, 2]];

 pic3 = Plot[Evaluate[{2 ca4[0, lL] + Sum[2 ca4[n, lL] Cos[(n Pi)/lL x], {n, 1, nn}] +
      Sum[0 Sin[(n Pi)/lL x], {n, 1, nn}]}], {x, -12, 14},
   PlotStyle → {{Thickness[0.003], Black}}, PlotRange → {{-4, 7}, {-.1, 1.1}}];
 Show[pic1, pic2, pic3, PlotRange → {{-3, 5}, {-.1, 1.1}},
  AspectRatio → Automatic, ImageSize → 800]]
```

Out[61]=



Since the periodic extension is continuous, the convergence is uniform in this case.

Example 6

In[62]:=
```
Clear[ff6, n, lL, nn];

ff6[x_] = x^2 Sign[x];
```

Since we already did E4 we do not need to calculate the Fourier Coefficients for this even function. We know that the cosine coefficients are 0 and the since coefficients are double the sine coefficients in E4.

In[64]:=
```
Module[{pic1, pic2, pic2a, pic3, lL, nn}, lL = 1;
 nn = 20;
 pic1 = Plot[{ff6[x]}, {x, -lL, lL}, PlotStyle → {{Thickness[0.007], Blue}}];

 pic2 = Plot[{fft[ff6[#] &, x, lL]}, {x, -5, 10},
   PlotStyle → {{Thickness[0.005], Green}}, Exclusions → Range[-11, 14, 2]];

 pic2a = Graphics[{
     {PointSize[0.015], Green,
      {Point[{#, -1}], Point[{#, 1}], Point[{#, 0}]} & /@ Range[-11, 13, 2]},
     {PointSize[0.01], White, {Point[{#, -1}], Point[{#, 1}]} & /@ Range[-11, 13, 2]}
    }];
 pic3 = Plot[Evaluate[{0 + Sum[0 Cos[(n Pi)/lL x], {n, 1, nn}] +
       Sum[2 cb4[n, lL] Sin[(n Pi)/lL x], {n, 1, nn}]}], {x, -12, 14},
   PlotStyle → {{Thickness[0.002], Black}}, PlotRange → {{-4, 7}, {-1.1, 1.1}}];
 Show[pic1, pic2, pic2a, pic3, PlotRange → {{-3, 5}, {-1.1, 1.1}},
  AspectRatio → Automatic, ImageSize → 800]]
```

Out[64]=